Lezione 30 (ore 60,61) - 3 dicembre 2025, 11:30 - 13:30 A13

(4.12) Teorema (convergenza del metodo TS(1)).

Siano t_0 un numero reale, F una funzione definita in R x R^n a valori in R^n e x_0 in R^n e si consideri il Problema di Cauchy:

(§) $x'(t) = F(t, x(t))$, $x(t_0) = x_0$, $t \in [t_0, t_f]$

Se tutte le derivate parziali prime di F(t,x) sono funzioni continue di t ed x e il Problema (§) ha una sola soluzione, *allora* per ogni $\lambda > 0$ il metodo TS(1) applicato al Problema (§) è *convergente per* $E \to 0$ e:

- N *tende a infinito come* $1 / \sqrt{E}$;
- *Per ogni* k: ET(k) *tende a zero come* $\sqrt{E}$.

(4.13) Realizzazione in *Scilab* (TS_1_pv).

```
function [T, X, PASSO] = TS_1_pv(x0, t0, tf, F, G2, E, LAMBDA, HMIN)
//
// Integra numericamente, sull'intervallo [t0,tf], il Problema
// di Cauchy in R(n):
//
// x' = F(t,x)
// x(t0) = x0
//
// con il metodo TS(1) - Eulero esplicito - a passo variabile.
//
// x0: condizione iniziale (colonna di n elementi)
// t0: istante iniziale (numero reale)
// tf: istante finale (numero reale)
// F: function che definisce l'equazione differenziale; F(t,x) deve
//    essere una colonna di n numeri reali
// G2: function che restituisce la derivata seconda in t della soluzione
//     dell'equazione differenziale che all'istante t assume valore x;
//     G2(t,x) deve essere una colonna di n numeri reali
// E: valore massimo della stima dell'errore locale (numero reale)
// LAMBDA: numero reale che stabilisce il valore massimo del passo
//         (OPZIONALE - valore predefinito: 1d-5)
// HMIN: valore minimo consentito del passo
//       (OPZIONALE - valore predefinito: (tf - t0) / 1d6)
//
// T = [t(0),...,t(N)]: riga contenente gli istanti di integrazione
// X = [x(0),...,x(N)]: matrice n x (N+1) contenente le approssimazioni
// PASSO = [h(0),...,h(N-1)]: riga contenente i passi di integrazione
//
// Valore degli argomenti opzionali
//
if ~exists('LAMBDA','l') then LAMBDA = 1d-5; end;
```

```
if ~exists('HMIN','l') then HMIN = (tf - t0) / 1d6; end;
//
// Inizializzazione delle variabili di uscita
//
T(1,1) = t0;
X(:,1) = x0;
PASSO = [];
//
// ciclo principale
//
while (T(1,$) < tf), // arresta la costruzione se ha raggiunto tf
  //
  // scelta del passo
  //
  Nd2x = norm(G2(T(1,$),X(:,$)));
  d = max(LAMBDA, Nd2x);
  PASSO(1,$+1) = min(sqrt(2*E/d), tf - T(1,$));
  //
  // calcolo approssimazione e nuovo istante di integrazione
  //
  X(:,$+1) = X(:,$) + F(T(1,$),X(:,$)) * PASSO(1,$);
  T(1,$+1) = T(1,$) + PASSO(1,$);
  //
  // arresta la costruzione se il passo calcolato risulta troppo
  // piccolo e non ha raggiunto tf
  //
  if (PASSO(1,$) < HMIN) & (T(1,$) < tf) then break; end;
  //
end;
//
// Verifica se l'integrazione ha raggiunto tf
//
if T(1,$) < tf then
   printf("\n\nIntegrazione interrotta a T = %3.2e", T(1,$));
end;
//
endfunction
```

(4.14) Esempio (svolto in classe il 4 dicembre).

Si consideri un pendolo realizzato da un punto pesante di massa m collegato da un filo inestensibile di lunghezza L ad un punto fisso. Supposto piano il moto del punto ed adottato l'angolo x tra la verticale discendente ed il filo, misurato in senso antiorario, come coordinata lagrangiana, l'equazione del moto risulta:

(ED) $$x''(t) = -\frac{g}{L}\,\mathrm{sen}\,x(t)$$

Per approssimare nell'intervallo $[t_0,t_f] = [0,3]$ s la soluzione del Problema di Cauchy che si ottiene considerando le condizioni iniziali:

(CI) $$x(0) = x_0 = \pi/4 \text{ rad} \quad , \quad x'(0) = 0$$

si utilizza, in *Scilab*, la procedura TS_1_pv. L'uso della procedura richiede:

- *La determinazione di un sistema di due equazioni differenziali di ordine uno equivalente all'equazione* (ED). Introdotte le variabili $u_1(t) = x(t)$, $u_2(t) = x'(t)$ si ottiene:

  (ED') $$u_1'(t) = u_2(t) \quad , \quad u_2'(t) = -\frac{g}{L}\,\text{sen}\,u_1(t)$$

  che si completa con le condizioni iniziali:

  (CI') $$u_1(0) = x_0 \quad , \quad u_2(0) = 0$$

- *La scrittura della* function *che definisce il sistema* (ED'):

```
function y = F(t,u)

   y = [                 u(2)  ;
         - (g/L) * sin( u(1) ) ];

endfunction
```

- *La determinazione della funzione che*, dati t ed u, *restituisce il valore della derivata seconda*, calcolata in t, *della soluzione del sistema* (ED') che passa per u all'istante t:

$$u''(t) = \begin{bmatrix} u_2'(t) \\ -(g/L)\,u_1'(t)\cos(u_1(t)) \end{bmatrix} = \begin{bmatrix} -(g/L)\,\text{sen}(u_1(t)) \\ -(g/L)\,u_2(t)\cos(u_1(t)) \end{bmatrix}$$

  e la scrittura della relativa function:

```
function y = G2(t,u)

   y = [       - (g/L) * sin( u(1) );
         - (g/L) * u(2) * cos( u(1) ) ];

endfunction
```

- *L'assegnamento dell'istante finale* t_f (s):

```
tf = 3;
```

- *L'assegnamento della colonna delle condizioni iniziali* (CI'):

```
u0 = [x0;0];
```

- *L'assegnamento del valore ai parametri*:

```
g = 9.82; // m/s^2
L = 1; // m
m = 1; // kg
```

- *La scelta del valore massimo consentito per la stima dell'errore locale*, E.

  Per ottenere un valore di E adeguato, occorre un criterio per giudicare l'accuratezza dell'approssimazione ottenuta dalla procedura. Per il sistema fisico in esame possiamo procedere come segue.

  (A) Considerato che durante il moto l'*energia meccanica*:

  $$EN(x(t)) = mgL(1-\cos x_1(t)) + \frac{1}{2} mL^2 (x_2(t))^2$$

  assume valore costante e pari al valore $EN(t_0)$ assunto all'istante t_0, come misura *relativa* dell'accuratezza dell'approssimazione possiamo scegliere *la variazione relativa dell'energia durante il moto approssimato*:

  $$Var\_EN = \frac{\max EN(u(t_k)) - \min EN(u(t_k))}{EN(u(t_0))}$$

  (B) Considerato che il moto del pendolo è periodico e che si ha:

  $$\min x_1(t) = - \max x_1(t) \quad \Rightarrow \quad \max x_1(t) + \min x_1(t) = 0$$

  come misura *relativa* dell'accuratezza dell'approssimazione possiamo scegliere *la variazione relativa dell'ampiezza dell'oscillazione durante il moto approssimato*:

  $$Var\_A = \frac{\max u_1(t_k) + \min u_1(t_k)}{u_1(t_0)}$$

  Questa scelta è ragionevole se l'intervallo $[t_0,t_f]$ include almeno una oscillazione della funzione $u_1(t_k)$.

  (C) Si ottiene la tabella che segue:

E	N	Var_EN (%)	Var_A (%)
10^{-3}	267	35.89	6.3
10^{-5}	2587	3.25	$5.99\ 10^{-1}$
10^{-7}	25779	0.32	$5.97\ 10^{-2}$

  Quale sia un valore di E adeguato dipende da quello che l'utilizzatore vuole ottenere. La tabelle suggerisce che al diminuire di E l'accuratezza dell'approssimazione aumenta.

(4.15) <u>Osservazione</u> (variazione di N e ET con E).

Siano N e M, rispettivamente, il numero di istanti di integrazione e il massimo valore di ET(k) ottenuto utilizzando la procedura TS_1_pv con E = $\underline{E}$ e N' e M' i corrispondenti valori ottenuti con E = $\alpha\,\underline{E}$. Per quanto detto nel Teorema (4.12) ci si aspetta che:

$$N'/N \approx 1/\alpha^{1/2} \quad \text{e} \quad M'/M \approx \alpha^{1/2}$$

Nella tabella finale dell'esempio precedente si ha $\alpha = 10^{-2}$, dunque ci si aspetta:

$$N'/N \approx 10 \quad \text{e} \quad M'/M \approx 1/10$$

La relazione riguardante l'aumento del numero di istanti di integrazione è evidentemente verificata:

$$2587/267 = 9.69 \quad \text{e} \quad 25779/2587 = 9.96$$

Non avendo possibilità di accedere all'errore totale, ci limitiamo a constatare che per la variazione relativa dell'energia si ha:

$$\text{Var\_EN}'/\text{Var\_EN} = 3.25/35.89 \approx 0.90\ 10^{-1} \quad \text{e} \quad 0.32/3.25 \approx 0.98\ 10^{-1}$$

e per la variazione relativa dell'ampiezza:

$$\text{Var\_A}'/\text{Var\_A} = 5.99\ 10^{-1}/6.3 \approx 0.95\ 10^{-1} \quad \text{e} \quad 5.97\ 10^{-2}/5.99\ 10^{-1} \approx 0.99\ 10^{-1}$$

(4.B) METODO TS(2)

(4.16) Ipotesi (regolarità delle soluzioni).

Supponiamo che *tutte* le soluzioni dell'equazione differenziale $x'(t) = F(t, x(t))$ abbiano *derivata terza continua*.

La richiesta è certamente soddisfatta se *tutte* le derivate parziali *seconde* della funzione $F(t,x)$ *esistono* e sono funzioni *continue* di t ed x.

(Infatti:

$$G_2(t,x) = \partial_t F(t,x) + \partial_x F(t,x) \cdot F(t,x)$$

ha derivate parziali prime continue e quindi:

$$G_3(t,x) = \partial_t G_2(t,x) + \partial_x G_2(t,x) \cdot F(t,x)$$

è continua. Allora, se $y(t)$ è soluzione dell'equazione differenziale:

$$y^{(3)}(t) = ((y'(t))')' = ((F(t,y(t)))')' = (G_2(t,y(t)))' = G_3(t,y(t))$$

è continua perché lo sono $G_3(t,x)$ ed $y(t)$.)

(4.17) Definizione (metodo TS(2)).

Il *metodo* TS(2) è definito dalle procedure seguenti.

- SCELTA di $h(k)$. Dati $E > 0$ e $\lambda > 0$, per ogni k si pone:

$$d(k) = \max \{ \ \lambda \ , \ \|y^{(3)}(t(k);\ x(k),t(k))\| \ \}$$

e poi:

$$h(k) = \min \{ \ \sqrt[3]{\frac{6E}{d(k)}} \ , \ tf - t(k) \ \}$$

- CALCOLO di x(k+1). Dopo aver scelto h(k) si pone:

$$x(k+1) = x(k) + F(\ t(k),x(k)\ )\ h(k) + \frac{1}{2}\ G_2(\ t(k),x(k)\ )\ h(k)^2$$

Il nome del metodo è conseguenza del fatto che la funzione di h utilizzata per il calcolo di x(k+1) si ottiene troncando al termine di ordine *due* la *serie di Taylor* di y(t(k) + h; x(k),t(k)) in h = 0.

(4.18) Osservazione (sulla scelta di h(k)).

Indicando con y(t) la soluzione y(t; x(k),t(k)) dell'equazione differenziale, per lo scostamento s(h) tra y(t(k) + h) e l'approssimazione calcolata dal metodo con un passo di lunghezza h a partire da (t(k),x(k)) si ha, utilizzando la Formula di Taylor in h = 0 con resto di Lagrange:

$$s(h) = -\ \frac{1}{6}\ y^{(3)}(t(k))\ h^3 + z(h)\ h^3 \qquad \text{con: } z(h) \to 0 \text{ per } h \to 0$$

Se $y^{(3)}(t(k))$ non è zero *allora*:

- *per* h *piccolo*: $-\ \frac{1}{6}\ y^{(3)}(t(k))\ h^3$ *è una buona stima di* s(h)
- si ha:

$$\left\| \ -\frac{1}{6}\ y^{(3)}(t(k))\ h^3 \ \right\| = E \quad \Leftrightarrow \quad h = \sqrt[3]{\frac{6E}{\|y^{(3)}(t(k))\|}}$$

Il parametro λ garantisce che:

$$\text{per ogni } k: \quad d(k) \geqslant \lambda \quad \text{e quindi} \quad h(k) \leqslant \sqrt[3]{\frac{6E}{\lambda}}$$

(4.19) Teorema (convergenza del metodo TS(2)).

Siano t_0 e $t_f > t_0$ numeri reali, F una funzione definita in R x R^n a valori in R^n, x_0 in R^n e si consideri il Problema di Cauchy:

(§) $\qquad x'(t) = F(\ t,x(t)\ ) \quad , \quad x(t_0) = x_0 \quad , \quad t \in [t_0,t_f]$

Se tutte le derivate parziali seconde di F(t,x) sono funzioni continue di t ed x e il Problema (§) ha una sola soluzione, *allora* per ogni $\lambda > 0$ il metodo TS(2) applicato al Problema (§) è *convergente per* $E \to 0$ e:

- N *tende a infinito come* $1 / \sqrt[3]{E}$;

- *Per ogni* k: ET(k) *tende a zero come* $\sqrt[3]{E^2} = E^{2/3}$

(4.20) <u>Osservazione</u>.

Si consideri il Problema di Cauchy (§). Per ogni $E > 0$, indichiamo con $N_1(E)$ e $ET_1(E)$ il numero di istanti di integrazione e l'errore totale massimo generati dal metodo TS(1) e con $N_2(E)$ e $ET_2(E)$ il numero di istanti di integrazione e l'errore totale massimo generati dal metodo TS(2). Per quanto detto nel Teorema (4.12) e nel Teorema (4.19), per $E \to 0$ si ha:

- $N_1(E) \,/\, N_2(E) \to +\infty$ come $1 / \sqrt[6]{E}$, dunque $N_1(E)$ *tende ad* ∞ *più rapidamente di* $N_2(E)$
- $ET_1(E) \,/\, ET_2(E) \to +\infty$ come $1 / \sqrt[6]{E}$, dunque $ET_2(E)$ *tende a* 0 *più rapidamente di* $ET_1(E)$

Ci si aspetta allora che, con lo stesso valore di E:

- TS(2) generi un *errore totale massimo più piccolo* di quello generato con TS(1)
- TS(2) raggiunga t_f con un *numero di passi inferiore* rispetto a TS(1)